Solution 22.10

(a) Here is a VBA program to implement the algorithm from Fig. 22.1*a*. It is set up to evaluate the integral in the problem statement,

```
Option Explicit
Sub TrapTest()
Dim a As Double, b As Double
Dim n As Integer
a = 0
b = 1
n = 4
MsgBox TrapEq(n, a, b)
End Sub
Function TrapEq(n, a, b)
Dim h As Double, x As Double, sum As Double
Dim i As Integer
h = (b - a) / n
x = a
sum = f(x)
For i = 1 To n - 1
  x = x + h
  sum = sum + 2 * f(x)
Next i
sum = sum + f(b)
TrapEq = (b - a) * sum / (2 * n)
End Function
Function f(x)
f = x ^ 0.1 * (1.2 - x) * (1 - Exp(20 * (x - 1)))
End Function
```

When the program is run, the result is



The percent relative error can be computed as

$$\varepsilon_t = \left|\frac{0.602298 - 0.478602}{0.602298}\right| \times 100\% = 20.54\%$$

Solution continued on the next page...

(b) Here is a VBA program to implement the algorithm from Fig. 22.1*b*. It is set up to evaluate the integral in the problem statement,

```
Option Explicit
Sub SimpTest()
Dim a As Double, b As Double
Dim n As Integer
a = 0: b = 1: n = 4
MsgBox SimpEq(n, a, b)
End Sub
Function SimpEq(n, a, b)
Dim h As Double, x As Double, sum As Double
Dim i As Integer
h = (b - a) / n
x = a
sum = f(x)
For i = 1 To n - 2 Step 2
  x = x + h
  sum = sum + 4 * f(x)
  x = x + h
  sum = sum + 2 * f(x)
Next i
x = x + h
sum = sum + 4 * f(x)
sum = sum + f(b)
SimpEq = (b - a) * sum / (3 * n)
End Function
Function f(x)
f = x ^ 0.1 * (1.2 - x) * (1 - Exp(20 * (x - 1)))
End Function
```

When the program is run, the result is



The percent relative error can be computed as

$$\varepsilon_t = \left|\frac{0.602298 - 0.529287}{0.602298}\right| \times 100\% = 12.12\%$$